
Comparing Structures

Structure comparison:

I More complicated than the simple integers
I Have to compare all the individual components
I Break down components recursively.



Comparing Structures. aless

Example
aless(X,Y) succeeds if

I X and Y stand for atoms and
I X is alphabetically less than Y.

aless(avocado, clergyman) succeeds.
aless(windmill, motorcar) fails.
aless(picture, picture) fails.



Comparing Structures. aless
Success Run out of characters in the first word:

aless(book, bookbinder).

Success The first character of the first word is
alphabetically less than one of the second:
aless(avocado, clergyman).

Recursion The first character is the same in both. Then have
to check the rest:
For aless(lazy, leather) check
aless(azy, eather).

Failure The first character of the first word is greater than
the first one of the second:
aless(book, apple).

Failure Reach the end of both words at the same time:
aless(apple, apple).

Failure Run out of characters for the second word:
aless(alphabetic, alp).



Representation

I Transform atoms into a recursive structure.
I List of integers (ASCII codes).
I Use built-in predicate atom_codes:

?- atom_codes(alp, [97,108,112]).
yes

?- atom_codes(alp, X).
X = [97,108,112] ?
yes

?-atom_codes(X, [97,108,112]).
X = alp ?
yes



First Task

Convert atoms to lists:

atom_codes(X, XL).
atom_codes(Y, YL).

Compare the lists:

alessx(XL, YL).

Putting together:

aless(X, Y) :-
atom_codes(X, XL),
atom_codes(Y, YL),
alessx(XL, YL).



Second Task

Compose alessx.
Success First word ends before second:

alessx([], [_|_]).

Success A character in the first is alphabetically less than
one in the second:
alessx([X|_], [Y|_]) :- X < Y.

Recursion The first character is the same in both. Then have
to check the rest:
alessx([H|X], [H|Y]) :- alessx(X, Y).

What about failing cases?



Program

aless(X, Y):-
atom_codes(X, XL),
atom_codes(Y, YL),
alessx(XL, YL).

alessx([], [_|_]).
alessx([X|_], [Y|_]):-

X < Y.
alessx([H|X], [H|Y]):-

alessx(X, Y).



Appending Two Lists

For any lists List1, List2, and List3
List2 appended to List1 is List3 iff either

I List1 is the empty list and List3 is List2, or
I List1 is a nonempty list and

I the head of List3 is the head of List1 and
I the tail of List3 is List2 appended to the tail of List1.

Program:

append([], L, L).
append([X|L1], L2, [X|L3]) :-

append(L1, L2, L3).



Using append

Test ?- append([a,b,c],[2,1],[a,b,c,2,1]).

Total List ?- append([a,b,c],[2,1],X).

Isolate ?- append(X,[2,1],[a,b,c,2,1]).
?- append([a,b,c],X,[a,b,c,2,1]).

Split ?- append(X,Y,[a,b,c,2,1]).



Inventory Example: Bicycle Factory

I To build a bicycle we need to know which parts to draw
from the supplies.

I Each part of a bicycle may have subparts.
I Task: Construct a tree-based database that will enable

users to ask questions about which parts are required to
build a part of bicycle.



Inventory Example: Bicycle Factory

I To build a bicycle we need to know which parts to draw
from the supplies.

I Each part of a bicycle may have subparts.
I Task: Construct a tree-based database that will enable

users to ask questions about which parts are required to
build a part of bicycle.



Parts of a Bicycle

I Basic parts:
basicpart(rim).
basicpart(spoke).
basicpart(rearframe).
basicpart(handles).

basicpart(gears).
basicpart(bolt).
basicpart(nut).
basicpart(fork).

I Assemblies, consisting of a quantity of basic parts or other
assemblies:

assembly(bike, [wheel,wheel,frame]).
assembly(wheel, [spoke,rim,hub]).
assembly(frame, [rearframe,frontframe]).
assembly(hub, [gears,axle]).
assembly(axle, [bolt,nut]).
assembly(frontframe, [fork,handles]).



Bike as a Tree

bike

wheel

spoke rim hub

gears axle

bolt nut

wheel

spoke rim hub

gears axle

bolt nut

frame

rearfr. frontfr.

fork handles



Program

Write a program that, given a part, will list all the basic parts
required to construct it.

Idea:
1. If the part is a basic part then nothing more is required.
2. If the part is an assembly, apply the same process (of

finding subparts) to each part of it.



Predicates: partsof

partsof(X, Y): Succeeds if X is a part of bike, and Y is the
list of basic parts required to construct X.

I Boundary condition. Basic part:
partsof(X, [X]) :- basicpart(X).

I Assembly:
partsof(X, P) :-

assembly(X, Subparts),
partsoflist(Subparts, P).

I Need to define partsoflist.



Predicates: partsoflist

I Boundary condition. List of parts for the empty list is
empty:
partsoflist([], []).

I Recursive case. For a nonempty list, first find partsof of
the head, then recursively call partsoflist on the tail of
the list, and glue the obtained lists together:
partsoflist([P|Tail], Total) :-

partsof(P, Headparts),
partsoflist(Tail, Tailparts),
append(Headparts, Tailparts, Total).

The same example using accumulators



Finding Parts

?- partsof(bike, Parts).

Parts=[spoke,rim,gears,bolt,nut,spoke,rim,
gears,bolt,nut,rearframe,fork,handles] ;

No

?- partsof(wheel, Parts).

Parts=[spoke, rim, gears, bolt, nut] ;
No



Using Intermediate Results

Frequent situation:

I Traverse a PROLOG structure.
I Calculate the result which depends on what was found in

the structure.
I At intermediate stages of the traversal there is an

intermediate value for the result.

Common technique:

I Use an argument of the predicate to represent the "answer
so far".

I This argument is called an accumulator.



Length of a List without Accumulators

Example
listlen(L, N) succeeds if the length of list L is N.

I Boundary condition. The empty list has length 0:
listlen([], 0).

I Recursive case. The length of a nonempty list is obtained
by adding one to the length of the tail of the list.

listlen([H|T], N) :-
listlen(T, N1),
N is N1 + 1.



Length of a List with an Accumulator

Example
lenacc(L, A, N) succeeds if the length of list L, when
added the number A, is N.

I Boundary condition. For the empty list, the length is
whatever has been accumulated so far, i.e. A:
lenacc([], A, A).

I Recursive case. For a nonempty list, add 1 to the
accumulated amount given by A, and recur to the tail of the
list with a new accumulator value A1:
lenacc([H|T], A, N) :-

A1 is A + 1,
lenacc(T, A1, N).



Length of a List with an Accumulator, Cont.

Example
Complete program:

listlenacc(L, N) :-
lenacc(L, 0, N).

lenacc([], A, A).
lenacc([H|T], A, N) :-

A1 is A + 1,
lenacc(T, A1, N).



Computing List Length

Example (Version without Accumulator)

listlen([a,b,c], N).
listlen([b,c], N1), N is N1 + 1.
listlen([c], N2), N1 is N2 + 1, N is N1 + 1.
listlen([], N3), N2 is N3 + 1, N1 is N2 + 1,
N is N1 + 1.
N2 is 0 + 1, N1 is N2 + 1, N is N1 + 1.
N1 is 1 + 1, N is N1 + 1.
N is 2 + 1.

N = 3

Computing List Length

Example (Version with an Accumulator)
listlenacc([a,b,c], N).

lenacc([a,b,c], 0, N).
A1 is 0+1, lenacc([b,c], A1, N).
lenacc([b,c], 1, N).
A2 is 1+1, lenacc([c], A2, N).
lenacc([c], 2, N).
A3 is 2+1, lenacc([], A3, N).
lenacc([], 3, N).

N = 3



Computing List Length

Example (Version with an Accumulator)
listlenacc([a,b,c], N).
lenacc([a,b,c], 0, N).

A1 is 0+1, lenacc([b,c], A1, N).
lenacc([b,c], 1, N).
A2 is 1+1, lenacc([c], A2, N).
lenacc([c], 2, N).
A3 is 2+1, lenacc([], A3, N).
lenacc([], 3, N).

N = 3



Computing List Length

Example (Version with an Accumulator)
listlenacc([a,b,c], N).
lenacc([a,b,c], 0, N).
A1 is 0+1, lenacc([b,c], A1, N).

lenacc([b,c], 1, N).
A2 is 1+1, lenacc([c], A2, N).
lenacc([c], 2, N).
A3 is 2+1, lenacc([], A3, N).
lenacc([], 3, N).

N = 3



Computing List Length

Example (Version with an Accumulator)
listlenacc([a,b,c], N).
lenacc([a,b,c], 0, N).
A1 is 0+1, lenacc([b,c], A1, N).
lenacc([b,c], 1, N).

A2 is 1+1, lenacc([c], A2, N).
lenacc([c], 2, N).
A3 is 2+1, lenacc([], A3, N).
lenacc([], 3, N).

N = 3



Computing List Length

Example (Version with an Accumulator)
listlenacc([a,b,c], N).
lenacc([a,b,c], 0, N).
A1 is 0+1, lenacc([b,c], A1, N).
lenacc([b,c], 1, N).
A2 is 1+1, lenacc([c], A2, N).

lenacc([c], 2, N).
A3 is 2+1, lenacc([], A3, N).
lenacc([], 3, N).

N = 3



Computing List Length

Example (Version with an Accumulator)
listlenacc([a,b,c], N).
lenacc([a,b,c], 0, N).
A1 is 0+1, lenacc([b,c], A1, N).
lenacc([b,c], 1, N).
A2 is 1+1, lenacc([c], A2, N).
lenacc([c], 2, N).

A3 is 2+1, lenacc([], A3, N).
lenacc([], 3, N).

N = 3



Computing List Length

Example (Version with an Accumulator)
listlenacc([a,b,c], N).
lenacc([a,b,c], 0, N).
A1 is 0+1, lenacc([b,c], A1, N).
lenacc([b,c], 1, N).
A2 is 1+1, lenacc([c], A2, N).
lenacc([c], 2, N).
A3 is 2+1, lenacc([], A3, N).

lenacc([], 3, N).

N = 3



Computing List Length

Example (Version with an Accumulator)
listlenacc([a,b,c], N).
lenacc([a,b,c], 0, N).
A1 is 0+1, lenacc([b,c], A1, N).
lenacc([b,c], 1, N).
A2 is 1+1, lenacc([c], A2, N).
lenacc([c], 2, N).
A3 is 2+1, lenacc([], A3, N).
lenacc([], 3, N).

N = 3



Computing List Length

Example (Version with an Accumulator)
listlenacc([a,b,c], N).
lenacc([a,b,c], 0, N).
A1 is 0+1, lenacc([b,c], A1, N).
lenacc([b,c], 1, N).
A2 is 1+1, lenacc([c], A2, N).
lenacc([c], 2, N).
A3 is 2+1, lenacc([], A3, N).
lenacc([], 3, N).

N = 3



List as an Accumulator

I Accumulators need not be integers.
I If a list is to be produced as a result, an accumulator will

hold a list produced so far.
I Wasteful joining of structures avoided.

Example (Reversing Lists)

reverse(List, Rev) :-
rev_acc(List, [], Rev).

rev_acc([], Acc, Acc).
rev_acc([X|T], Acc, Rev) :-

rev_acc(T, [X|Acc], Rev).



Bicycle Factory

Recall how parts of bike were found. Inventory example

partsoflist has to find the parts coming from the list
[wheel,wheel,frame]:

I Find parts of frame.
I Append them to [] to find parts of [frame].
I Find parts of wheel.
I Append them to the parts of [frame] to find parts of
[wheel,frame].

I Find parts of wheel.
I Append them to the parts of [wheel,frame] to find parts

of [wheel,wheel,frame].
Wasteful!



Bicycle Factory

Improvement idea: Get rid of append. Use accumulators.
partsacc(X, A, P): parts of X, when added to A, give P.

partsof(X, P) :- partsacc(X, [], P).

partsacc(X, A, [X|A]) :- basicpart(X).
partsacc(X, A, P) :-

assembly(X, Subparts),
partsacclist(Subparts, A, P).

partsacclist([], A, A).
partsacclist([P|Tail], A, Total) :-

partsacc(P, A, Headparts),
partsacclist(Tail, Headparts, Total).



Difference Structures

Compute parts of wheel without and with accumulator:

Example (Without Accumulator)
?- partsof(wheel, P).
X = [spoke, rim, gears, bolt, nut] ;
No

Example (With Accumulator)
?- partsof(wheel, P).
X = [nut, bolt, gears, rim, spoke] ;
No

Reversed order.



Difference Structures

How to avoid wasteful work and retain the original order at the
same time?

Difference structures.



Open Lists and Difference Lists

I Consider the list [a,b,c|Ho].
I The structure of the list is known up to a point.
I If, at some point, Ho is unbound then we have an open list.
I Informally, Ho is a called a “hole”.



Open Lists and Difference Lists

I Unify Ho with [d,e]:
?- List=[a,b,c|Ho], Ho=[d,e].

List=[a,b,c,d,e]

I We started with an open list and “filled” in the hole with the
structure.

Open Lists and Difference Lists

I The result of filling in the hole in an open list with a “proper”
list is a “proper” list.

I What happens if we instantiate the hole with an open list?

?- List=[a,b,c|Ho], Ho=[d,e|Y].

?- List=[a,b,c,d,e|Y].



Open Lists and Difference Lists

Open Lists and Difference Lists

I Filling in the hole with a proper list, again:
I ?- List=[a,b,c|Ho], Ho=[d,e].

I ?- List=[a,b,c,d,e].

I Is not it the same as append([a,b,c],[d,e],List)?



open_append

I We can define append in terms of “hole filling”.
I Assume the first list is given as an open list.
I Define a predicate that fills in the hole with the second list.
I A naive and limited way of doing this:
open_append([H1,H2,H3|Hole],L2):-Hole=L2.

?- List=[a,b,c|Ho], open_append(List,[d,e]).

List=[a,b,c,d,e]

Ho=[d,e]

I Improvement is needed: This version assumes having a
list with three elements and the hole.

Improvement Idea

I One often wants to say about open lists something like
“take the open list and fill in the hole with ...”

I Hence, one should know both an open list and a hole.
I Idea for list representation: Represent a list as an open list

together with the hole.
I Such a representation is called a difference list.
I Example: The difference list representation of the list
[a,b,c] is the pair of terms [a,b,c|X] and X.

Difference Lists

I Introduce a notation for difference lists.

I Idea: We are usually interested the open list part of difference
list, without the hole.

I From the pair [a,b,c|Ho] and Ho we are interested in
[a,b,c].

I “Subtracting” the hole Ho from the open list [a,b,c|Ho].

I [a,b,c|Ho]-Ho.

I The - has no interpreted meaning. Instead one could define any
operator to use there.

diff_append. Version 2

I diff_append(OpenList-Hole, L2) :- Hole=L2.

?- DList=[a,b,c|Ho]-Ho,
diff_append(DList,[d,e]).

DList=[a,b,c,d,e]-[d,e]
Ho=[d,e]

I Has to be improved again: We are not interested in the “filled
hole” in the instantiation of Ho hanging around.

diff_append. Version 3

I Let diff_append return the open list part of the first argument:

diff_append(OpenList-Hole, L2, OpenList) :-
Hole=L2.

?- DList=[a,b,c|Ho]-Ho,
diff_append(Dlist,[d,e],Ans).

Dlist=[a,b,c,d,e]-[d,e]
Ho=[d,e]
Ans=[a,b,c,d,e]

I It is better now. Ans looks as we would like to.

I Still, there is a room for improvement: The diff_append

I takes a difference list as its first argument,
I a proper list as its second argument, and
I returns a proper list.

I Let’s make it more uniform.

diff_append. Version 3
I Better, but not the final approximation: diff_append takes two

difference lists and returns an open list:

diff_append(
OpenList1-Hole1, OpenList2-Hole2, OpenList1

) :-
Hole1=OpenList2.

?- Dlist=[a,b,c|Ho]-Ho,
diff_append(Dlist,[d,e|Ho1]-Ho1,Ans).

Dlist=[a,b,c,d,e|Ho1]-[d,e|Ho1]
Ho=[d,e|Ho1]
Ans=[a,b,c,d,e|Ho1]

I We have returned an open list but we want a difference list.

I The first list has gained the hole of the second list.

I All we need to ensure is that we return the hole of the second
list.

diff_append. Version 4

diff_append can be made more compact:

diff_append(
OpenList1-Hole1,
Hole1-Hole2,
OpenList1-Hole2

).



diff_append. Usage

I Add an element at the end of a list:
add_to_back(L-H, El, Ans) :-

diff_append(L-H, [El|H1]-H1, Ans-[]).

?- add_to_back([a,b,c|H]-H, e, Ans).

H = [e]

Ans = [a,b,c,e]

Difference Structures

Both accumulators and difference structures use two
arguments to build the output structure.

Assumulators: the “result so far” and the “final result”.
Difference structures: the (current approximation of the) “final

result” and the “hole in there where the further
information can be put”.



Bicycle Factory

Use holes.

partsof(X, P) :-
partshole(X, P-Hole),
Hole=[].

partshole(X, [X|Hole]-Hole) :-
basicpart(X).

partshole(X, P-Hole) :-
assembly(X, Subparts),
partsholelist(Subparts, P-Hole).

partsholelist([], Hole-Hole).
partsholelist([P|Tail], Total-Hole) :-

partshole(P, Total-Hole1),
partsholelist(Tail, Hole1-Hole).



Bicycle Factory. Detailed View

partsof(X, P) :-
partshole(X, P-Hole),
Hole=[].

I partshole(X, P-Hole) builds the result in the second
argument P and returns in Hole a variable.

I Since partsof calls partshole only once, it is
necessary to terminate the difference list by instantiating
Hole with []. (Filling the hole.)

I Alternative definition of partsof:
partsof(X, P) :- partshole(X, P-[]).
It ensures that the very last hole is filled with [] even
before the list is constructed.



Bicycle Factory. Detailed View

partshole(X, [X|Hole]-Hole) :-
basicpart(X).

I It returns a difference list containing the object (basic part)
in the first argument.

I The hole remains open for further instantiations.



Bicycle Factory. Detailed View

partshole(X, P-Hole):-
assembly(X, Subparts),
partsholelist(Subparts, P-Hole).

I Finds the list of subparts.
I Delegates the traversal of the list to partsholelist.
I The difference list P-Holeis passed to partsholelist.



Bicycle Factory. Detailed View

partsholelist([P|Tail], Total-Hole) :-
partshole(P, Total-Hole1),
partsholelist(Tail, Hole1-Hole).

I partshole starts building the Total list, partially filling it
with the parts of P, and leaving a hole Hole1 in it.

I partsholelist is called recursively on the Tail. It
constructs the list Hole1 partially, leaving a hole Hole in it.

I Since Hole1 is shared between partshole and
partsholelist, after getting instantiated in
partsholelist it gets also instantiated in partshole.

I Therefore, at the end Total consists of the portion that
partshole constructed, the portion of Hole1
partsholelist constructed, and the hole Hole.

